
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc163397897]Corrections required
1.1 [bookmark: _Toc163397898]f_AT_ResponseCheck
	Functionname
	f_AT_ResponseCheck

	Reason for change
	Testcase will not stop even error occurred when using function f_AT_ResponseCheck.

	Summary of change
	Add self.kill at the end of the function

	TTCN module
	NAS_AuxiliaryDefsAndFunctions.ttcn

	MCC160 Comment
	This function is used several times in a test case to check the same and different AT responses. If the test stops after the first failure, the results of the later checks will not be known.  
Alternative solution implemented: returned the result of the match and stop the test case, if any fail, once all checks have been performed.



Before Change:
	function f_AT_ResponseCheck(charstring p_AT_Response,
                              charstring p_ExpectedString,
                              charstring p_Text := "",
                              charstring p_Expression := "*[""]([^""]+)[""]*",
                              integer p_Group := 0)
  { /* p_Expression := "*[""]([^""]+)[""]*";    TTCN-3 regular expression: anything + quote + no quote (at least one character) + quote + anything;
     *                                          => the text between the most inner quotes is returned */
    var charstring v_Expression := p_Expression;
    var charstring v_Result;
    var template charstring v_ExpectedString := pattern p_ExpectedString;
    var boolean v_MatchResult;
    var charstring v_CRLF := oct2char('0D'O) & oct2char('0A'O);
    
    v_Result := regexp(p_AT_Response, v_Expression, p_Group);
    
    v_MatchResult := match(v_Result, v_ExpectedString);
    
    if (v_MatchResult == true) {
      f_SetVerdict(pass,__FILE__, __LINE__, p_Text);
    } else  {
      f_ErrorLog(__FILE__, __LINE__, v_CRLF & "Value expected: " & p_ExpectedString & v_CRLF & "Value received: " & v_Result);  // if anything went wrong, show it
      f_SetVerdict(fail,__FILE__, __LINE__, p_Text);	  
	      }
  }



After Change:
	function f_AT_ResponseCheck(charstring p_AT_Response,
                              charstring p_ExpectedString,
                              charstring p_Text := "",
                              charstring p_Expression := "*[""]([^""]+)[""]*",
                              integer p_Group := 0)
  { /* p_Expression := "*[""]([^""]+)[""]*";    TTCN-3 regular expression: anything + quote + no quote (at least one character) + quote + anything;
     *                                          => the text between the most inner quotes is returned */
    var charstring v_Expression := p_Expression;
    var charstring v_Result;
    var template charstring v_ExpectedString := pattern p_ExpectedString;
    var boolean v_MatchResult;
    var charstring v_CRLF := oct2char('0D'O) & oct2char('0A'O);
    
    v_Result := regexp(p_AT_Response, v_Expression, p_Group);
    
    v_MatchResult := match(v_Result, v_ExpectedString);
    
    if (v_MatchResult == true) {
      f_SetVerdict(pass,__FILE__, __LINE__, p_Text);
    } else  {
      f_ErrorLog(__FILE__, __LINE__, v_CRLF & "Value expected: " & p_ExpectedString & v_CRLF & "Value received: " & v_Result);  // if anything went wrong, show it
      f_SetVerdict(fail,__FILE__, __LINE__, p_Text);	  
	  self.kill;    // => test case will be terminated immediately
    }
  }


MCC160 Implementation:
 var boolean v_Result;    

...

 //@siclog "Steps 14" siclog@
    v_Result := f_AT_ResponseCheck(v_AT_Response,
                       v_CurrentYearStrExt& "/12/31,14:[0-9]#(2):[0-9]#(2)", // TZ GMT+1, dst (1 char) ignored, check date and hour, ignore mins, secs @sic R5s200903, R5-221451, R5s220502 sic@
                       "Step 14 (UT and local time zone)",
                       v_Expression_CTZR,
                       2);
    v_Result := v_Result and f_AT_ResponseCheck(v_AT_Response,
                       "1", // DST=1
                       "Step 14  DST",
                       v_Expression_CTZR,
                       1);
    v_AT_Response := f_UT_ChkOperatorName (UT, LONG_NAME);
    v_Result := v_Result and f_AT_ResponseCheck (v_AT_Response,
                        v_FullNameString,
                        "Step 14 (long network name)");
    v_AT_Response := f_UT_ChkOperatorName (UT, SHORT_NAME);
    v_Result := v_Result and f_AT_ResponseCheck (v_AT_Response,
                        v_ShortNameString,
                        "Step 14 (short network name)");
    v_AT_Response := f_UT_ChkTimeZone (UT);
    v_Result := v_Result and f_AT_ResponseCheck(v_AT_Response,
                       "*[0-9]#(2)/[0-9]#(2)/[0-9]#(2),[0-9]#(2):[0-9]#(2):[0-9]#(2)\+04*", // only TZ to be checked
                       "Step 14 (local time zone)");

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 14");
...


  function f_AT_ResponseCheck(charstring p_AT_Response,
                              charstring p_ExpectedString,
                              charstring p_Text := "",
                              charstring p_Expression := "*[""]([^""]+)[""]*",
                              integer p_Group := 0) return boolean
  { /* p_Expression := "*[""]([^""]+)[""]*";    TTCN-3 regular expression: anything + quote + no quote (at least one character) + quote + anything;
     *                                          => the text between the most inner quotes is returned */
    var charstring v_Expression := p_Expression;
    var charstring v_Result;
    var template charstring v_ExpectedString := pattern p_ExpectedString;
    var boolean v_MatchResult;
    var charstring v_CRLF := oct2char('0D'O) & oct2char('0A'O);
    
    v_Result := regexp(p_AT_Response, v_Expression, p_Group);
    
    v_MatchResult := match(v_Result, v_ExpectedString);
    
    if (v_MatchResult == true) {
      f_SetVerdict(pass,__FILE__, __LINE__, p_Text);
    } else  {
      f_ErrorLog(__FILE__, __LINE__, v_CRLF & "Value expected: " & p_ExpectedString & v_CRLF & "Value received: " & v_Result);  // if anything went wrong, show it
      f_SetVerdict(fail,__FILE__, __LINE__, p_Text);
    }
    return v_MatchResult;
  }

